Q:

Volume of a region of revolution

I'm given the region bounded by $x=0$ and $x=4y^2-y^3$.
Show[
 ContourPlot[{x == 4 y^2 - y^3, x == 0}, {x, -1, 10}, {y, 0, 4}],
 RegionPlot[0 <= x <= 4 y^2 - y^3, {x, 0, 256/27}, {y, 0, 4}],
 Axes -> True
 ]

I'm asked to rotate it about the x-axis.
Show[
 RevolutionPlot3D[{4 y^2 - y^3, y, 0}, {y, 0, 4}, 
  RevolutionAxis -> {1, 0, 0}, AxesLabel -> {"x", "y"}],
 RevolutionPlot3D[{0, y, 0}, {y, 0, 4}, RevolutionAxis -> {1, 0, 0}],
 BaseStyle -> Opacity[0.4]
 ]

Now, this is similar to a question I posted at Area of surface of revolution. I've found the volume using the cylindrical shell method.
Integrate[2 \[Pi] y (4 y^2 - y^3), {y, 0, 4}]

The answer is $512\pi/5$. I'd like to use Mathematica's Volume command to find the volume, but I've been unsuccessful. Tough problem. Does anyone have a suggestion?
Update: This can help show that it is a rotation about the x-axis.
Manipulate[
 ParametricPlot3D[{4 y^2 - y^3, y Cos[t], y Sin[t]}, {y, 0, 4}, {t, 0,
    tau}, AxesLabel -> {"x", "y"}, PerformanceGoal -> "Quality", 
  PlotStyle -> Opacity[0.8], PlotRange -> {{0, 10}, {-4, 4}, {-4, 4}}],
 {{tau, 0.1}, 0, 2 \[Pi]}]

A:

ir = ImplicitRegion[
  x^2 + y^2 < 16 && 
   0 < z < 4 (x^2 + y^2) - (x^2 + y^2)^(3/2), {{x, -4, 4}, {y, -4, 
    4}, {z, 0, 10}}]
Volume[ir]

yields: $512\pi/5$

